

**Description of the fungicide treatments used in the strawberry leaf blight
experiments over 5 years**

Year	Treatment	Rate (/acre)^a	Timing^b
1996	1. Captec 4L	3 qt	Every 10-14 days starting after renovation
	2. Thiram 65W	5 lb	
	3. Syllit 65W	2 lb	
	4. Benlate 50W	16 oz	
	5. Benlate 50W + Captec 4L	8 oz + 3 qt	
	6. Benlate 50W + Thiram 65W	8 oz + 5 lb	
	7. Nova 40W	5 oz	
	8. Nova 40W + Captec 4L	5 oz + 3 qt	
	9. Nova 40W + Thiram 65W	5 oz + 5 lb	
	10. Bravo 720	3 pt	
	11. Calcium cholride	4.5 lb	
	12. Untreated control	B	B
1997	1. Captec 4L	3 qt	5, 15, 26 May; 9 June
	2. Thiram 65W	5 lb	
	3. Benlate 50W	16 oz	
	4. Syllit 65W	2 lb	
	5. Benlate 50W + Captec 4L	8 oz + 3 qt	
	6. Benlate 50W + Thiram 65W	8 oz + 5 lb	
	7. Nova 40W	5 oz	
	8. Untreated control	B	B
1998	1. Benlate 50W + Thiram 65W	8 oz + 5 lb	4, 14, 22 May; 6 June
	2. Captec 4L	3 qt	
	3. Syllit 65W	2 lb	
	4. Benlate 50W + Captec 4L	8 oz + 3 qt	
	5. Abound 2.08F	11 oz	
	6. Benlate 50W	16 oz	
	7. Nova 40W	5 oz	
	8. Thiram 65W	5 lb	
	9. Untreated control	B	B
1999	1. Captec 4L	3 qt	14, 21, 28 May; 4, 11 June
	2. Captec 4L + Benlate 50W	2 qt + 1 lb	
	3. Thiram 65W + Benlate 50W	4 lb + 1 lb	
	4. Switch 62.5WG	14 oz	
	5. Flint 50WG/Switch 62.5 WG ^c	3 oz / 14 oz	

	6. Elevate 50 WDG	1.5 lb	
	7. Abound 2.08F	11 fl oz	
	8. Rovral 50 WP	1.5 lb	
	9. Thiram 65 WP	5 lb	
	10. Untreated control	B	B
2000	1. Switch 62.5 WG	11 oz	12, 19, 26,30 May; 7, 20 June
	2. Switch 62.5 WG /	11 oz	12, 19 May
	Elevate 50WG	1.5 lb	26, 30 May; 7,20 June
	3. Flint 50WG /	3 oz	12, 19 May
	Switch 62.5 WG	11 oz	26, 30 May; 7, 20 June
	4. Switch 62.5 WG /	11 oz	12, 19 May
	Flint 50 WG /	3 oz	26 May
	Switch 62.5 WG	11 oz	30 May; 7, 20 June
	5. Elevate 50 WG	1.5 lb	
	6. Captec 4L	3 qt	
	7. Captec 4L + Benlate 50W	3 qt + 1 lb	
	8. Abound 2.08F	11 fl oz	
	9. Nova 40W	4 oz	
	10. Untreated control	B	B

^aRates are given in English units because this is how the labels are written.

^bIf description is blank, application timing is the same as treatment #1 in the given year.

^cFungicides were alternated.

Input SAS program to perform the generalized linear mixed model analysis.

```

/* Example of analysis of Phomopsis leaf blight for 1997      */
/* using %GLIMMIX.                                           */
/* trt: treatment; rep: block; su: sampling unit within plot; */
/* x: number of diseased leaflets; n: leaflet number         */
/* Read in data... Data strung across lines                  */

data phomop;
input trt rep su x n @@;
datalines;
8 1 1 4 15 8 1 2 12 15 8 1 3 11 15 8 1 4 12 15
8 1 5 9 15 1 2 1 9 15 1 2 2 4 15 1 2 3 3 15
1 2 4 9 15 1 2 5 7 15 2 3 1 3 15 2 3 2 9 15
2 3 3 4 15 2 3 4 4 15 2 3 5 6 15 7 4 1 4 15
7 4 2 0 15 7 4 3 1 15 7 4 4 2 15 7 4 5 0 15
6 1 1 2 15 6 1 2 2 15 6 1 3 1 15 6 1 4 0 15
6 1 5 6 15 3 2 1 1 15 3 2 2 1 15 3 2 3 1 15
3 2 4 0 15 3 2 5 2 15 5 3 1 0 15 5 3 2 1 15
5 3 3 1 15 5 3 4 0 15 5 3 5 0 15 8 4 1 6 15
8 4 2 1 15 8 4 3 9 15 8 4 4 4 15 8 4 5 8 15
4 1 1 5 15 4 1 2 7 15 4 1 3 5 15 4 1 4 0 15
4 1 5 5 15 5 2 1 0 15 5 2 2 0 15 5 2 3 0 15
5 2 4 1 15 5 2 5 0 15 8 3 1 12 15 8 3 2 5 15
8 3 3 6 15 8 3 4 9 15 8 3 5 5 15 1 4 1 6 15
1 4 2 11 15 1 4 3 1 15 1 4 4 2 15 1 4 5 6 15
1 1 1 0 15 1 1 2 7 15 1 1 3 0 15 1 1 4 2 15
1 1 5 4 15 8 2 1 9 15 8 2 2 7 15 8 2 3 6 15
8 2 4 0 15 8 2 5 4 15 6 3 1 1 15 6 3 2 0 15
6 3 3 0 15 6 3 4 0 15 6 3 5 1 15 4 4 1 1 15
4 4 2 0 15 4 4 3 0 15 4 4 4 0 15 4 4 5 0 15
3 1 1 4 15 3 1 2 4 15 3 1 3 0 15 3 1 4 6 15
3 1 5 5 15 4 2 1 5 15 4 2 2 3 15 4 2 3 3 15
4 2 4 3 15 4 2 5 0 15 3 3 1 3 15 3 3 2 0 15
3 3 3 0 15 3 3 4 3 15 3 3 5 8 15 6 4 1 0 15
6 4 2 0 15 6 4 3 2 15 6 4 4 1 15 6 4 5 2 15
2 1 1 0 15 2 1 2 5 15 2 1 3 1 15 2 1 4 0 15
2 1 5 6 15 2 2 1 6 15 2 2 2 2 15 2 2 3 0 15
2 2 4 6 15 2 2 5 7 15 7 3 1 0 15 7 3 2 0 15
7 3 3 0 15 7 3 4 0 15 7 3 5 0 15 5 4 1 0 15
5 4 2 1 15 5 4 3 2 15 5 4 4 0 15 5 4 5 1 15
5 1 1 0 15 5 1 2 1 15 5 1 3 1 15 5 1 4 1 15
5 1 5 0 15 6 2 1 0 15 6 2 2 3 15 6 2 3 2 15
6 2 4 1 15 6 2 5 1 15 4 3 1 1 15 4 3 2 6 15
4 3 3 0 15 4 3 4 0 15 4 3 5 1 15 3 4 1 0 15
3 4 2 0 15 3 4 3 2 15 3 4 4 3 15 3 4 5 1 15
7 1 1 0 15 7 1 2 0 15 7 1 3 0 15 7 1 4 0 15
7 1 5 0 15 7 2 1 1 15 7 2 2 1 15 7 2 3 0 15
7 2 4 0 15 7 2 5 0 15 1 3 1 3 15 1 3 2 5 15
1 3 3 7 15 1 3 4 2 15 1 3 5 1 15 2 4 1 1 15
2 4 2 0 15 2 4 3 8 15 2 4 4 5 15 2 4 5 11 15
;
/* The following statement tells SAS where to find %glimmix */
/* (you may have it in a different directory)               */
/* Note: the macro may have several examples attached.      */
/* Remove these examples and re-save the macro.              */

%include 'c:\Program Files\Sas Institute\SAS\V8\stat\sasmacro\glimmix.sas';

/* Model A                                                  */
/* See Piepho (1999) for a nice description of the SAS input. */
/* In addition to the REPL analysis, least squares means    */
/* (estimated linear predictors) for TRT are calculated, as well */

```

```

/* as pairwise differences (PDIFF) */
/* Note: COVTEST does a large-sample test for all estimated */
/* variance terms (probably not valid with typical sample sizes) */
/* Fixed effects are given on the MODEL statement; random effects */
/* are given on the RANDOM statements, and residual effects are */
/* are given with the REPEATED statement (the latter only when */
/* non-default options are needed (see Model C)). */
/* There are several ways of specifying the random terms; here the */
/* experimental error is given with the trt*rep subject (sub=) */
/* designation; >int= is intercept. */

%glimmix(data =phomop, procopt=covtest,
          stmts=%str(title 'MODEL A, 1997';
                    class trt rep su ;
                    model x/n = trt / solution chisq;
                    random rep;
                    random int / sub=trt*rep;
                    lsmeans trt / cl pdiff;
                    ),
          error=binomial, maxit=50,
          link=logit
);

/* Model B */
/* Need to specify initial estimates of variances in PARMs for: */
/* rep, rep*trt, sampling error, and residual (phi). */
/* Four variances in total. The fourth one is equated to 1 here */
/* (with the EQCONS option) */

%glimmix(data =phomop, procopt=covtest,
          stmts=%str(title 'MODEL B, 1997';
                    class trt rep su ;
                    model x/n = trt / solution chisq;
                    random rep;
                    random int su/ sub=trt*rep;
                    parms (0.04) (0.03) (0.4) (1) /eqcons=4;
                    lsmeans trt / cl pdiff;
                    ),
          error=binomial, maxit=50,
          link=logit
);

/* Model C */
/* The REPEATED statement is used to define the residual */
/* variance (not needed in A, because in A one is using the */
/* default of a single phi). */
/* Only needed when there are special conditions. */

%glimmix(data =phomop, procopt=covtest,
          stmts=%str(title 'MODEL C, 1997';
                    class trt rep su ;
                    model x/n = trt / solution chisq;
                    random rep;
                    random int / sub=trt*rep;
                    repeated su / sub=rep group=trt;
                    lsmeans trt / cl pdiff;
                    ),
          error=binomial, maxit=50,
          link=logit
);

/* Next run is just to get predicted p_i's without random effects; */
/* use as first step in 2-step estimation of local phi (residual). */
/* Otherwise, ignore output */

```

```

/* Data file is created (>results=) to be used in the next step          */

%glimmix(data =phomop, procopt=covtest,
          stmts=%str(title 'first step to get p_hats for later analysis';
          class trt rep su ;
          model x/n = trt / solution chisq;
          ),
          error=binomial,
          link=logit,
          out=results
);

/* Create needed variable (PQT) from above output                        */
/* to use as input for Model D, below                                   */
/* (in convention of glimmix, mu is p_hat, and _wght is n)                */

data phomop2;
  set results;
  pqt=log(mu*(1-mu)/_wght);

/* Model D                                                                */
/* Use original data and the predicted mu's (p_hats) from                */
/* a fixed-effects-only model.                                           */
/* See the REPEATED with LOCAL option for the                            */
/* binary power-law.                                                      */
/* Note: overdispersion scale (phi) is A of the binary power-law.        */
/* Thus, this calculated scale has no physical meaning at all            */
/* (because b' and p_hat must be considered).                            */

%glimmix(data =phomop2, procopt=covtest,
          stmts=%str(title 'MODEL D, 1997 (uses REPEATED/LOCAL option)';
          class trt rep su ;
          model x/n = trt / solution chisq;
          random rep;
          random int / sub=trt*rep;
          repeated / local = exp(pqt);
          lsmeans trt / cl pdiff;
          ),
          error=binomial,
          link=logit, maxit=50
);

/* Model E                                                                */
/* One must change the PARMs statement to match the                     */
/* number of variance terms. One needs (in this order):                 */
/* an initial estimate for rep, rep*trt, variance for each               */
/* treatment level, and then the residual phi (=1).                      */
/* With 1997, there are 8 treatment levels, so one specifies             */
/* 1+1+8+1 = 11 variance estimates.                                       */
/* One also must indicate that the last term is always 1.                */

%glimmix(data =phomop, procopt=covtest,
          stmts=%str(title 'MODEL E, 1997';
          class trt rep su ;
          model x/n = trt / solution chisq;
          random rep;
          random int / sub=trt*rep;
          random su / sub=rep group=trt;
          parms (0.01) (0.6) (.1) (.1) (.1) (.1) (.1)
                (.1) (.1) (.1) (1) /eqcons=11;
          lsmeans trt / cl pdiff;
          ),

```

```
    error=binomial, maxit=50,  
    link=logit  
);  
  
run;
```

Example of truncated and annotated output of the %GLIMMIX procedure for Model A with the 1997 data set^a

```

MODEL A, 1997

The Mixed Procedure

Model Information
Class Level Information

Class      Levels      Values
trt         8      1 2 3 4 5 6 7 8
rep         4      1 2 3 4
su          5      1 2 3 4 5

```

Below are the estimates of the variance (or >covariance= parameters). For this Model, there is a variance for rep (or block), experimental error (= Intercept with subject=trt*rep), and residual (phi; overdispersion term). For other models, this section of output would list the additional estimated variance terms (e.g., sampling variance [Model B]; b= of binary power law [Model D])^a.

Note: for this data set, block variance estimate (called rep here) is 0. The standard error of the estimated variance term is based on large-sample-theory for linear mixed models. The Z value is the estimate divided by the standard error. For small data sets, and for GLMMs, it is unlikely that the estimated variance terms have a normal distribution. Use the Z value to be a very rough guide about the distance of the estimate from 0.

```

Convergence criteria met.

Covariance Parameter Estimates

Cov Parm      Subject      Estimate      Standard      Z
                        Error      Value      Pr Z
rep
Intercept      trt*rep      0.2195      0.1436      1.53      0.0633
Residual                        2.2021      0.2739      8.04      <.0001

```

The following Fit Statistics are only directly useful for a linear mixed model (not a GLMM).

The log likelihood is not for the (overdispersed) binomial, but for a normal distribution at the last call of MIXED. The actual quasi-likelihood for overdispersed proportion data is incorporated into the Deviance (later).

```

Fit Statistics

-2 Res Log Likelihood      529.3
AIC (smaller is better)    533.3
AICC (smaller is better)   533.4
BIC (smaller is better)    532.1

```

The following results are the estimated parameters for treatment (tau) effects. There is an intercept (mean), and coefficients for each treatment level (see eq. 1 in paper). The linear predictors are derived from these values. For example, the estimated linear predictor for treatment 2 (conditional on the expected values of the random effects) is: $-0.1491 + (-0.8069) = -0.9560$.

```

Solution for Fixed Effects

```

Effect	trt	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		-0.1491	0.2917	3	-0.51	0.6444
trt	1	-0.7306	0.4195	21	-1.74	0.0962
trt	2	-0.8069	0.4206	21	-1.92	0.0688
trt	3	-1.6431	0.4478	21	-3.67	0.0014
trt	4	-1.6408	0.4482	21	-3.66	0.0015
trt	5	-3.2219	0.6067	21	-5.31	<.0001
trt	6	-2.2667	0.4872	21	-4.65	0.0001
trt	7	-3.3533	0.6285	21	-5.34	<.0001
trt	8	0

The following are the tests for treatment effect (see SAS for a description of >Type 3' nomenclature). Although both chi-square and F-tests can be performed, we use only the F-test.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
trt	7	21	60.86	8.69	<.0001	<.0001

The least squares means are the estimated linear predictors for each treatment level (at the expected values of the random effects). Also printed (among other things): estimated standard errors of the least square means, degrees of freedom $= (8-1)*(4-1)$, t-statistic for least squares means, and 95% confidence intervals. An estimate of p for the 1st treatment can be calculated as $1/(1+\exp(-(-0.8798))) = 0.293$. The standard error of this estimate of p requires a more complicated calculation (see Littell et al. [23]).

Least Squares Means

Effect	trt	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
trt	1	-0.8798	0.3016	21	-2.92	0.0082	0.05	-1.5069	-0.2526
trt	2	-0.9560	0.3031	21	-3.15	0.0048	0.05	-1.5863	-0.3257
trt	3	-1.7923	0.3398	21	-5.28	<.0001	0.05	-2.4989	-1.0857
trt	4	-1.7900	0.3403	21	-5.26	<.0001	0.05	-2.4976	-1.0824
trt	5	-3.3711	0.5320	21	-6.34	<.0001	0.05	-4.4774	-2.2647
trt	6	-2.4158	0.3903	21	-6.19	<.0001	0.05	-3.2274	-1.6042
trt	7	-3.5025	0.5568	21	-6.29	<.0001	0.05	-4.6603	-2.3446
trt	8	-0.1491	0.2917	21	-0.51	0.6145	0.05	-0.7557	0.4574

The following are pair-wise differences of estimated linear predictors (for all possible pairings of treatment levels). For example, the difference between treatments 1 and 2 is: $-0.8798 - (-0.9560) = 0.0762$, given as Estimate below. The standard error is the standard error of the difference (SED). T-value is the results of a t-test for the difference. Note: the difference of estimated p=s cannot be determined by transforming the difference of the estimated linear predictors; one must first estimate the p=s, then determine the differences. However, the differences of the estimated linear predictors are appropriate for determining significant differences between treatments.

Differences of Least Squares Means

Effect	trt	_trt	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
trt	1	2	0.07623	0.4276	21	0.18	0.8602	0.05	-0.8129	0.9654
trt	1	3	0.9125	0.4543	21	2.01	0.0576	0.05	-0.03223	1.8573
trt	1	4	0.9102	0.4547	21	2.00	0.0584	0.05	-0.03530	1.8557
trt	1	5	2.4913	0.6115	21	4.07	0.0005	0.05	1.2195	3.7630
trt	1	6	1.5360	0.4932	21	3.11	0.0052	0.05	0.5103	2.5617

trt	1	7	2.6227	0.6332	21	4.14	0.0005	0.05	1.3059	3.9395
trt	1	8	-0.7306	0.4195	21	-1.74	0.0962	0.05	-1.6031	0.1419
trt	2	3	0.8363	0.4553	21	1.84	0.0804	0.05	-0.1106	1.7831
trt	2	4	0.8340	0.4557	21	1.83	0.0815	0.05	-0.1137	1.7816

..

Results cut here for brevity (there are $8 \times 7/2 = 28$ pairs for the 8 treatment).

..

GLMM statistics. Deviance is based on the quasi-likelihood (conditional on the random effects). Extra-Dispersion Scale is the same as the overdispersion or Residual variance ($\hat{\sigma}$; see above). Scaled Deviance is Deviance/Scale. Pearson chi-square is another type of goodness of fit. Dividing Deviance by the degrees of freedom gives another way of estimating extra-dispersion scale. There are 160 observations and 8 fixed-effects parameters in this data set; thus, $v = 152$. This gives $D/v = 2.35$ for another measure of overdispersion.

GLIMMIX Model Statistics

Description	Value
Deviance	358.6275
Scaled Deviance	162.8542
Pearson Chi-Square	311.4997
Scaled Pearson Chi-Square	141.4533
Extra-Dispersion Scale	2.2021

^aAnnotations are given in Times Roman. The SAS output is in **Courier**.